


# Відео гра

**Завдання:** Створити відео гру, в якій необхідно вибирати маркер правильного кольору на екрані комп'ютера у відповідності до кольору світлодіода на платі mCore.



## Що будемо вивчати

1. Блоки, пов'язані з відео.

## Крок 1. Підготовка

1. Створіть новий проект (File-> New).
2. Видаліть існуючий спрайт (клік правою кнопкою миші, "delete").
3. Додайте чотири нові спрайти у вигляді різнокольорових маркерів – зеленого, червоного, жовтого та синього кольорів.

## Крок 2. Програмування кольору світлодіода на платі mCore

1. Створіть на сцені скрипт ввімкнення камери і встановлення прозорості.



2. Додайте до основного скрипта на сцені сигнал початку гри.

```

repeat 3
  play tone on note C4 beat Half
  wait 0.3 secs
  play tone on note 0 beat 0
  wait 0.3 secs
play tone on note F4 beat Half
wait 1.2 secs
play tone on note 0 beat 0

```

3. Створіть змінні “Score”, “Answer” і “Time”. Змінна “Score” відповідає за рахунок, “Answer” визначає чи зроблено гравцем вибір кольору, “Time” – таймер гри. Визначте початкові значення змінних в основному скрипті на сцені.

```

set Score to 0
set Answer to 0
set Time to 60

```

4. Створіть змінну “Color”, яка відповідає за колір світлодіода, визначеного програмою.

5. Додайте до основного скрипта на сцені команди випадкового визначення кольору світлодіода на платі mCore.


```

forever
  if Answer = 0 then
    set Color to pick random 1 to 4
    set led on board all red 0 green 0 blue 0
    if Color = 1 then
      set led on board led left red 0 green 20 blue 0
    if Color = 2 then
      set led on board led left red 20 green 0 blue 0
    if Color = 3 then
      set led on board led left red 20 green 20 blue 0
    if Color = 4 then
      set led on board led left red 0 green 0 blue 20
    set Answer to 1

```


### Крок 3. Вибір кольору маркера на екрані комп'ютера

1. Створіть скрипти для маркерів, які будуть постійно перевіряти чи співпадає їх колір з кольором світлодіода на платі mCore і можливість вибору на екрані комп'ютера. Якщо доторкнутись до правильного маркера, то він зникає на 0,2 секунди. Зверніть увагу, що значення кольору для кожного маркера має бути власне і співпадати з значенням відповідного кольору світлодіода!



```
when green flag clicked
  forever loop
    show
    if Color = 1 then
      if video motion on this sprite > 30 then
        hide
        set Answer to 0
        change Score by 1
        broadcast Touched
        wait 0.2 secs
```

2. Створіть на сцені скрипт сигналу правильного вибору маркера.



```
when I receive Touched
  play tone on note C4 beat Half
  wait 1 secs
  play tone on note 0 beat 0
```

### Крок 4. Кінець гри

1. Створіть на сцені скрипт, який визначає кінець гри, після закінчення 60 секунд.



```
when green flag clicked
  wait 3 secs
  repeat until Time = 0
    wait 1 secs
    change Time by -1
  play tone on note E5 beat Half
  wait 0.3 secs
  play tone on note C4 beat Half
  wait 1 secs
  play tone on note 0 beat 0
  set video transparency to 100 %
  stop all
```